Chapter 4
Finite Dimensional Dynamics of Evolutionary Equations with Maple

Alexei G. Kushner and Ruslan I. Matviichuk

4.1 Introduction

The theory of finite dimensional dynamics is a natural development of the theory of dynamical systems. Dynamics make it possible to find families that depend on a finite number of parameters among all solutions of evolutionary differential equations.

The basic ideas and methods of this theory were formulated in [7, 13]. In the same papers, finite dynamics were constructed for the Kolmogorov–Petrovsky–Piskunov and the Korteweg–de Vries equations.

Second-order dynamics of the Burgers–Huxley equation were constructed in [10].

Dynamics of third order were found for equations of the Rapoport–Leas type arising in the theory of two-phase filtration. These dynamics were used for constructing attractors [2, 3].

The paper is devoted to the finite dimensional dynamics of some evolution equations that arise in physics, mathematical biology, and mathematical economics. Among them are the Fisher–Kolmogorov–Petrovsky–Piskunov [21] equation and its generalization and the Black–Scholes equation [4].

When finding dynamics, we have to carry out calculations in jet spaces. This leads to cumbersome formulas. To facilitate calculations and avoid mistakes, we use the packages DifferentialGeometry and JetCalculus of the system of symbolic

A. G. Kushner (✉)
Moscow Pedagogical State University, Moscow, Russia
e-mail: kushner@physics.msu.ru

R. I. Matviichuk
Lomonosov Moscow State University, Moscow, Russia

M. Ulan, E. Schneider (eds.), *Differential Geometry, Differential Equations, and Mathematical Physics*, Tutorials, Schools, and Workshops in the Mathematical Sciences, https://doi.org/10.1007/978-3-030-63253-3_4

calculations Maple. A description of the basics of working with these packages can be found in [20].

Examples of dynamics calculations are considered and the program codes are given. These codes, with minor modifications, can be used to compute dynamics and find exact or approximate solutions to other evolutionary equations.

The structure of this paper is as follows.

In the first two sections, we give basic definitions and describe methods of the theory. The details can be found in [5, 7, 11, 13].

In the third section, we calculate first- and second-order dynamics of the Fisher–Kolmogorov–Petrovsky–Piskunov (FKPP) equation

$$u_t = u_{xx} + f(u)$$

and apply them to construct approximate solutions.

Note that in the considered example, the dynamics equation is solved exactly, but the group of shifts along the evolutionary vector field was found only approximately.

In the fourth section, we consider the reaction–diffusion equation with a convection term. This equation differs from the FKPP equation by the presence of a first-order derivative with respect to x (see [16, 18, 19]):

$$u_t + H(u)_x = u_{xx} + f(u).$$

It is proved that this equation has first-order dynamics for any smooth functions H and f, but second-order dynamics exist only when the function H is quadratic and the function f is cubic (see Theorem 4.3).

The fifth section is devoted to the Black–Scholes equation that came from mathematical finance theory. We construct two series of its exact solutions.

Some Maple files can be found on the website, d-omega.org.

4.2 Symmetries of ODEs

Any ordinary differential equation

$$y^{(k+1)} = h\left(x, y, y', y'', \ldots, y^{(k)}\right) \tag{4.1}$$

can be considered as a one-dimensional distribution P on the jet space $J^k(\mathbb{R})$. This distribution is generated by the vector field

$$\mathcal{D} = \frac{\partial}{\partial x} + y_1 \frac{\partial}{\partial y_0} + \cdots + y_k \frac{\partial}{\partial y_{k-1}} + h \frac{\partial}{\partial y_k}.$$

Here, $x, y_0, y_1, \ldots, y_k$ are canonical coordinates on $J^k(\mathbb{R})$ [9].

Integral curves of P are prolongations of trajectories of Eq. (4.1) into the space $J^k(\mathbb{R})$.

Definition 4.1 A vector field X on $J^k(\mathbb{R})$ is called an *infinitesimal symmetry* of Eq. (4.1) if translations along X preserve P.

All infinitesimal symmetries form the Lie algebra with respect to the Lie bracket. We denote this algebra by $\mathsf{Symm}\,\mathsf{P}$.

Definition 4.2 An infinitesimal symmetry is called *characteristic* if translations along it preserve each integral curve of the distribution P.

Characteristic symmetries form an ideal in $\mathsf{Symm}\,\mathsf{P}$, which we denote by $\mathsf{Char}\,\mathsf{P}$.

Definition 4.3 The quotient Lie algebra

$$\mathsf{Shuff}\,\mathsf{P} := \mathsf{Symm}\,\mathsf{P}\,/\,\mathsf{Char}\,\mathsf{P}$$

is called the Lie algebra of *shuffling* symmetries.

Each shuffling symmetry can be identified with a vector field of the form

$$S_\phi = \phi\frac{\partial}{\partial y_0} + \mathcal{D}(\phi)\frac{\partial}{\partial y_1} + \mathcal{D}^2(\phi)\frac{\partial}{\partial y_2} + \cdots + \mathcal{D}^k(\phi)\frac{\partial}{\partial y_k},$$

where ϕ is a function on $J^k(\mathbb{R})$ that is called a *generating function* of the corresponding shuffling symmetry.

If the function h does not depend on x, then y_1 is a generating function of a symmetry of Eq. (4.1).

4.3 Flows on ODE's Solution Spaces

Consider the following evolutionary partial differential equation:

$$\frac{\partial u}{\partial t} = \phi\left(x, u, \frac{\partial u}{\partial x}, \ldots, \frac{\partial^k u}{\partial x^k}\right). \tag{4.2}$$

Let $\phi = \phi(x, y_0, \ldots, y_k)$ be a generating function of some shuffling symmetry of Eq. (4.1), and let Φ_t be the translation along the vector field S_ϕ from $t = 0$ to t. Let $L_{y(x)} = \{y_0 = y(x)\}$ be the graph of some solution $y = y(x)$ of Eq. (4.1), and let

$$L_{y(x)}^{(k)} = \{y_0 = y(x), y_1 = y'(x), \ldots, y_k = y^{(k)}(x)\} \tag{4.3}$$

be its prolongation into the space $J^k(\mathbb{R})$. Shifting the curve $L^{(k)}_{y(x)}$ along the trajectories of the vector field S_ϕ, we get the surface

$$\Phi_t\left(L^{(k)}_{y(x)}\right) \subset J^k(\mathbb{R}^2)$$

that is a prolongation of the graph of a solution of evolution equation (4.2). Here, $J^k(\mathbb{R}^2)$ is the k-jet space of functions with two independent variables t and x. Describe two methods for constructing solutions of equation (4.2).

Method 1 The space of solutions of equation (4.1) can be identified with the space $\mathbb{R}^{k+1}$ by indicating the initial data of solutions at a fixed point $x = x_0$. Then, the shift transformation Φ_t defines the transformation of the space $\mathbb{R}^{k+1}$ with coordinates $y_0, y_1, \ldots, y_k$. Therefore, we can consider transformations of this space instead of transforming curves. Such transformations are given by shifts $\overline{\Phi}_t$ along the vector field

$$E_\phi = \overline{\phi}\frac{\partial}{\partial y_0} + \mathcal{D}(\overline{\phi})\frac{\partial}{\partial y_1} + \mathcal{D}^2(\overline{\phi})\frac{\partial}{\partial y_2} + \cdots + \mathcal{D}^k(\overline{\phi})\frac{\partial}{\partial y_k},$$

where $\overline{\phi}$ is a restriction of the function ϕ to Eq. (4.1).

Let $y = y(x; \mathbf{a})$ be the solution of equation (4.1) with initial data

$$y(x_0) = a_0,\, y'(x_0) = a_1, \ldots, y^{(k)}(x_0) = a_k.$$

Applying the transformation $\overline{\Phi}_t$ to the point $\mathbf{a} = (a_0, \ldots, a_k)$, we obtain a one-parameter family $y(x; \overline{\Phi}_t(\mathbf{a}))$ of solutions of equation (4.1). Then, the function

$$u(t, x) = y(x, \overline{\Phi}_t(\mathbf{a}))$$

is a solution of the evolutionary Eq. (4.2) with the initial data $u(0, x) = y(x; \mathbf{a})$.

Method 2 The transformation Φ_t acting on the jet space $J^k(\mathbb{R})$ generates a transformation Φ_t^* acting on functions. Let Φ_t^{-1} be the inverse transformation for Φ_t. Curve (4.3) is generated by the system of equalities

$$y_0 - y(x) = 0,\; y_1 - y'(x) = 0, \ldots,\; y_k - y^{(k)}(x) = 0. \tag{4.4}$$

Applying the transformation $\left(\Phi_t^{-1}\right)^*$ to (4.4), we obtain the following systems:

$$\Psi^0(t, x, y_0, \ldots, y_k) = 0, \quad \Psi^1(t, x, y_0, \ldots, y_k) = 0, \ldots, \Psi^k(t, x, y_0, \ldots, y_k) = 0.$$

Solving it with respect to $y_0, \ldots, y_k$, we find a coordinate representation of the curve $\Phi_t\left(L^{(k)}_{y(x)}\right)$:

$$y_0 = Y_0(t, x), y_1 = Y_1(t, x), \dots, y_k = Y_k(t, x). \tag{4.5}$$

The function $u(t, x) = Y_0(t, x)$ is a solution of equation (4.2). The remaining functions in (4.5) correspond to the partial derivatives:

$$\frac{\partial^j u}{\partial x^j} = Y_j(t, x), \quad j = 1, \dots, k.$$

The first method is convenient when the solution of equation (4.1) or the shift transformation Φ_t can be found only approximately. The second method, on the contrary, is applicable in the case when the solution and shift transformation can be found explicitly.

Definition 4.4 If ϕ is a generating function of a shuffling symmetry of Eq. (4.1), then Eq. (4.1) is called a (finite dimensional) *dynamics* of Eq. (4.2). The number $k + 1$ is called the *order* of the dynamics.

Thus, an evolutionary equation determines a flow on the solution space of an ordinary differential equation.

The following theorem (see [2]) provides a method for calculating finite dimensional dynamics of evolutionary equations.

Theorem 4.1 *The ordinary differential equation*

$$F = y_{k+1} - h(x, y_0, y_1, \dots, y_k) = 0$$

is a dynamics of evolutionary equation (4.2) if and only if

$$[\phi, F] = 0 \operatorname{mod} \mathbf{DF}, \tag{4.6}$$

where $\mathbf{DF} = \langle F, D(F), D^2(F), \dots \rangle$ *is the differential ideal generated by the function* F*,*

$$D = \frac{\partial}{\partial x} + y_1 \frac{\partial}{\partial y_0} + y_2 \frac{\partial}{\partial y_1} + \cdots$$

is the operator of total derivative, and

$$[\phi, F] = \sum_{i \geqslant 0} \left(\frac{\partial \phi}{\partial y_i} D^i(F) - \frac{\partial F}{\partial y_i} D^i(\phi) \right)$$

is the Poisson–Lie bracket.

The Poisson–Lie bracket is a prolongation of the classical Poisson bracket into the jet space (see, for example, [11]). Note that the Poisson–Lie bracket is skew-symmetric $\mathbb{R}$-bilinear and satisfies the Jacobi identity. From the skew-symmetric property, it follows that $[\phi, \phi] = 0$, and therefore the equation $\phi = 0$ is a dynamics of Eq. (4.2).

4.4 The Fisher–Kolmogorov–Petrovsky–Piskunov Equation

The equation

$$u_t = u_{xx} + f(u) \tag{4.7}$$

is known as the Fisher–Kolmogorov–Petrovsky–Piskunov equation or the reaction–diffusion equation.

It describes processes of heat and mass transfer, propagation of a dominant gene [6, 8], propagation of flame [21], reaction–diffusion [14], and ferroelectric domain wall motion in an electric field [17]. For example, Eq. (4.7) with

$$f(u) = (1 - u^2)(m - u),$$

$-1 < m \leqslant 0$, describes active transmission of an electric impulse in neuron, and it is known as Nagumo's equation [15].

B. Kruglikov and O. Lychagina [7] presented an analysis of finite dimensional dynamics of Eq. (4.7).

4.4.1 Second-Order Dynamics

Equation (4.7) admits second-order dynamics if the function $f(u)$ is cubic (see [7]):

$$f(u) = f_3 u^3 + f_2 u^2 + f_1 u + f_0,$$

where $f_0, \dots, f_3 \in \mathbb{R}$. Then,

$$\phi = y_2 + f_3 y_0^3 + f_2 y_0^2 + f_1 y_0 + f_0.$$

Find second-order dynamics in the form of the Liénard equation [12], i.e. put

$$F := y_2 - A(y_0) y_1 - B(y_0), \tag{4.8}$$

where A and B are some smooth functions. Consider two cases.

Case 1: $f_3 > 0$ Then, we can put $f_3 = 2q^2$. The restriction of the Poisson–Lie bracket to dynamics (4.8) gives us the following system of equations:

$$\begin{cases} fB' - Bf' = 0, \\ (2B - f)A' = 0, \\ A'' = 0, \\ B'' + 2AA' + 12q^2 y_0 + 2f_2 = 0. \end{cases}$$

Solving this system, we get

$$A(y_0) = A_1 y_0 + A_0$$

and

$$B(y_0) = -\frac{1}{3}(A_1^2 + 6q^2)y_0^3 - (A_0 A_1 + f_2)y_0^2 + B_1 y_0 + B_0,$$

where A_0, A_1, B_0, and B_1 are constants that we find from the first two equations: A_0 is arbitrary, and

$$A_1 = 0, \quad B_0 = -f_0, \quad B_1 = -f_1.$$

Then, dynamics (4.8) has the form

$$F = y_2 + 2q^2 y_0^3 + f_2 y_0^2 - A_0 y_1 + f_1 y_0 + f_0 = \phi - A_0 y_1. \tag{4.9}$$

Therefore, the restriction ϕ to equation $F = 0$ is

$$\overline{\phi} = A_0 y_1.$$

Case 2: $f_3 < 0$ We can put $f_3 = -2q^2$, and we get the equation

$$B'' + 2AA' - 12q^2 y_0 + 2f_2 = 0$$

instead of the last equation in system (4.4.1). Then, $A(y_0) = A_1 y_0 + A_0$ and

$$B(y_0) = -\frac{1}{3}(A_1^2 - 6q^2)y_0^3 - (A_0 A_1 + f_2)y_0^2 + B_1 y_0 + B_0,$$

and we get three solutions:

1. A_0 is arbitrary, $A_1 = 0$, $B_0 = -f_0$, $B_1 = -f_1$;
2. $A_0 = -\frac{f_2}{2q}$, $A_1 = 3q$, $B_0 = \frac{f_0}{2}$, $B_1 = \frac{f_1}{2}$;
3. $A_0 = \frac{f_2}{2q}$, $A_1 = -3q$, $B_0 = \frac{f_0}{2}$, $B_1 = \frac{f_1}{2}$.

So, we get the following dynamics:

$$F_1 = y_2 - A_0 y_1 - 2q^2 y_0^3 + f_2 y_0^2 + f_1 y_0 + f_0, \tag{4.10}$$

$$F_2 = y_2 - \left(3q y_0 - \frac{f_2}{2q}\right) y_1 + q^2 y_0^3 - \frac{f_2}{2} y_0^2 - \frac{f_1}{2} y_0 - \frac{f_0}{2}, \tag{4.11}$$

$$F_3 = y_2 + \left(3q y_0 - \frac{f_2}{2q}\right) y_1 + q^2 y_0^3 - \frac{f_2}{2} y_0^2 - \frac{f_1}{2} y_0 - \frac{f_0}{2}. \tag{4.12}$$

The restrictions ϕ to this dynamics are

$$\overline{\phi}_1 = A_0 y_1,$$

$$\overline{\phi}_2 = \frac{1}{2q}\left(-6q^3 y_0^3 + 6q^2 y_1 y_0 + 3q(f_2 y_0^2 + f_1 y_0 + f_0) - f_2 y_1\right),$$

$$\overline{\phi}_3 = \frac{1}{2q}\left(-6q^3 y_0^3 - 6q^2 y_1 y_0 + 3q(f_2 y_0^2 + f_1 y_0 + f_0) + f_2 y_1\right),$$

respectively.

As a result, we obtain the following theorem.

Theorem 4.2 *The FKPP equation*

$$u_t = u_{xx} + f_3 u^3 + f_2 u^2 + f_1 u + f_0, \tag{4.13}$$

with nonzero f_3, admits the following second-order dynamics of the form:

$$F = y_2 - A(y_0) y_1 - B(y_0).$$

- *If $f_3 > 0$, i.e. $f_3 = 2q^2$, then the dynamics has form (4.9);*
- *If $f_3 < 0$, i.e. $f_3 = -2q^2$, then the dynamics have forms (4.10)–(4.12).*

Here, q is a nonzero number.

4.4.2 Integration of the Dynamics

Consider, for example, dynamics (4.12). Corresponding differential equation has the form

$$y'' + \left(3q y - \frac{f_2}{2q}\right) y' + q^2 y^3 - \frac{f_2}{2} y^2 - \frac{f_1}{2} y - \frac{f_0}{2} = 0. \tag{4.14}$$

The distribution P is generated by the vector field

$$\mathcal{D} = \frac{\partial}{\partial x} + y_1\frac{\partial}{\partial y_0} - \left(\left(3qy_0 - \frac{f_2}{2q}\right)y_1 + q^2y_0^3 - \frac{f_2}{2}y_0^2 - \frac{f_1}{2}y_0 - \frac{f_0}{2}\right)\frac{\partial}{\partial y_1} \tag{4.15}$$

or by the differential 1-forms

$$\omega_1 = dy_0 - y_1dx,$$
$$\omega_2 = dy_1 + \left(\left(3qy_0 - \frac{f_2}{2q}\right)y_1 + q^2y_0^3 - \frac{f_2}{2}y_0^2 - \frac{f_1}{2}y_0 - \frac{f_0}{2}\right)dx.$$

This distribution has two shuffling symmetries:

$$\begin{aligned}S_1 =& \overline{\phi}\frac{\partial}{\partial y_0} + \mathcal{D}(\overline{\phi})\frac{\partial}{\partial y_1}\\ =& \left(-3q^2y_0^3 + \frac{3}{2}f_2y_0^2 + \left(-3qy_1 + \frac{3}{2}f_1\right)y_0 + \frac{1}{2q}f_2y_1 + \frac{3}{2}f_0\right)\frac{\partial}{\partial y_0}\\ &\left(3q^3y_0^4 - 2qf2y_0^3 + \frac{1}{4q^2}(-6q^3f_1 + qf_2^2)y_0^2 + \frac{1}{4q^2}(qf_1f_2 - 6q^3f_0)y_0\right.\\ &\left.-3qy_1^2 + \frac{1}{4q^2}(6f_1q^2 + f_2^2)y_1 + \frac{1}{4q}f_0f_2\right)\frac{\partial}{\partial y_1}\end{aligned}$$

and

$$\begin{aligned}S_2 =& S_{y_1} = y_1\frac{\partial}{\partial y_0} + \mathcal{D}(y_1)\frac{\partial}{\partial y_1}\\ =& y_1\frac{\partial}{\partial y_0} + \left(-q^2y_0^3 + \frac{1}{2}f_2y_0^2 + \left(-3qy_1 + \frac{1}{2}f_1\right)y_0 + \frac{1}{2q}f_2y_1 + \frac{1}{2}f_0\right)\frac{\partial}{\partial y_1}.\end{aligned}$$

The vector fields S_1 and S_2 define commutative symmetry Lie algebra:

$$[S_1, S_2] = 0.$$

According to the Lie–Bianchi theorem [5, 11], the ordinary differential equation $F = 0$ is integrable by quadratures. In order to construct its first integrals, we construct two differential 1-forms ϖ_1 and ϖ_2 instead of the forms ω_1 and ω_2. We choose them so that they form a dual basis for the vector fields S_1 and S_2, i.e. $\varpi_i(S_j) = \delta_{ij}$, where δ_{ij} is the Kronecker delta. Compose the matrix

$$W = \begin{Vmatrix}\omega_1(S_1) & \omega_1(S_2)\\ \omega_2(S_1) & \omega_2(S_2)\end{Vmatrix}.$$

Determinant of the matrix is

$$\det W = \frac{1}{4q}\Big(12q^5y_0^6 + 36y_1q^4y_0^4 - 12y_0^2(y_0^3f_2 + f_1y_0^2 + f_0y_0 - 3y_1^2)q^3$$
$$- 18\left(-\frac{2}{3}y_1^2 + f_1y_0^2 + f_0y_0 + \frac{4}{3}y_0^3f_2\right)q^2y_1$$
$$+ \Big(3f_2^2y_0^4 + 6f_1y_0^3f_2 + (3f_1^2 + 6f_0f_2)y_0^2 + (6f_0f_1 - 12y_1^2f_2)y_0$$
$$+3f_0^2 - 6y_1^2f_1)q + 3y_1f_2(f_2y_0^2 + f_1y_0 + f_0)\Big).$$

In the domain of the plane (y_0, y_1) where $\det W \neq 0$, there exists the inverse matrix W^{-1}. Define differential 1-forms ϖ_1 and ϖ_2:

$$\begin{Vmatrix}\varpi_1\\ \varpi_2\end{Vmatrix} = W^{-1}\begin{Vmatrix}\omega_1\\ \omega_2\end{Vmatrix}.$$

Since the Lie bracket $[S_1, S_2] = 0$, we get

$$d\varpi_i\,(S_1, S_2) = S_1(\varpi_i(S_2)) - S_2(\varpi_i(S_1)) - \varpi([S_1, S_2]) = 0.$$

This means that the forms ϖ_1 and ϖ_2 are closed. Due to the Poincaré lemma, there exist functions H_1 and H_2 such that $\varpi_1 = dH_1$ and $\varpi_2 = dH_2$. These functions are first integrals of the ordinary differential equation $F = 0$. Integrating the forms ϖ_1 and ϖ_2 along an arbitrary path in the space $J^1(\mathbb{R})$, we find these integrals. We do not write them for general case because of their bulkiness.

4.4.3 Construction Solutions of the FKPP Equation by Dynamics

To construct solutions of equation (4.13), we use Method 1 (see page 126).

Let $y(x; \mathbf{a})$, where $\mathbf{a} = (a_0, a_1)$, be the solution of ordinary differential equation (4.14) with initial conditions $y(x_0) = a_0$ and $y'(x_0) = a_1$. Let Φ_t be the shift transformation along the vector field S_1 from $t = 0$ to t. Since Φ_t is a symmetry of Eq. (4.14), the function $y(x; \Phi_t(\mathbf{a}))$ is a solution of this equation too.

The transformation Φ_t is defined by the solution of the ordinary equations

$$\begin{cases} \dfrac{dy_0}{dt} = -3q^2y_0^3 + \dfrac{3}{2}f_2y_0^2 + \left(-3qy_1 + \dfrac{3}{2}f_1\right)y_0 + \dfrac{1}{2q}f_2y_1 + \dfrac{3}{2}f_0, \\ \dfrac{dy_1}{dt} = 3q^3y_0^4 - 2qf2y_0^3 + \dfrac{1}{4q^2}(-6q^3f_1 + qf_2^2)y_0^2 \\ + \dfrac{1}{4q^2}(qf_1f_2 - 6q^3f_0)y_0 - 3qy_1^2 + \dfrac{1}{4q^2}(6f_1q^2 + f_2^2)y_1 + \dfrac{1}{4q}f_0f_2 \end{cases} \tag{4.16}$$

with initial conditions $y_0(0) = y_0$ and $y_1(0) = y_1$.

Therefore, if we manage to solve this system and find the flow of the vector field S_1 in explicit form, then we can construct an exact solution of the FKPP equation. Otherwise, we can use numerical methods to system (4.16). As a result, we obtain approximate solutions of equation (4.13).

Example 4.1 Consider the equation

$$u_t = u_{xx} - 2u^3 + 1. \tag{4.17}$$

Then, $\phi = y_2 - 2y_0^3 + 1$, and we have three dynamics:

$$F_1 = y_2 + 3y_0y_1 + y_0^3 - \frac{1}{2}; \tag{4.18}$$

$$F_2 = y_2 - 3y_0y_1 + y_0^3 - \frac{1}{2}; \tag{4.19}$$

$$F_3 = y_2 - \alpha y_1 - 2y_0^3 + 1, \tag{4.20}$$

where α is a constant.

Consider dynamics (4.18), for example, i.e. suppose that $F = F_1$. The restriction of the function ϕ to the equation $F = 0$ is

$$\overline{\phi} = -3y_0^3 - 3y_0y_1 + \frac{3}{2}.$$

The distribution P is generated by the differential 1-forms

$$\omega_1 = dy_0 - y_1dx, \tag{4.21}$$

$$\omega_2 = dy_1 + \left(3y_0y_1 + y_0^3 - \frac{1}{2}\right)dx. \tag{4.22}$$

The vector fields of shuffling symmetries are

$$S_1 = -\left(3y_0^3 + 3y_0y_1 - \frac{3}{2}\right)\frac{\partial}{\partial y_0} + \left(-3y_1^2 + 3y_0^4 - \frac{3}{2}y_0\right)\frac{\partial}{\partial y_1}, \tag{4.23}$$

Fig. 4.1 The vector field S_1



$$S_2 = y_1 \frac{\partial}{\partial y_0} - \left(3y_0 y_1 + y_0^3 - \frac{1}{2}\right) \frac{\partial}{\partial y_1}. \tag{4.24}$$

Remark 4.1 The vector field S_1 has a stable focus at the point $y_0 = \dfrac{1}{\sqrt[3]{2}}, y_1 = 0$ (see Fig. 4.1).

The matrix W is

$$W = \begin{Vmatrix} -3y_0^3 - 3y_1 y_0 + \dfrac{3}{2} & y_1 \\ 3y_0^4 - 3y_1^2 - \dfrac{3}{2} y_0 & \dfrac{1}{2} - 3y_1 y_0 - y_0^3 \end{Vmatrix}.$$

It is nondegenerate if

$$\det W = 9y_0^2 y_1^2 - 3y_0^3 + 9y_0^4 y_1 + 3y_0^6 - \frac{9}{2} y_1 y_0 + 3y_1^3 + \frac{3}{4} \neq 0.$$

The differential 1-forms ϖ_1 and ϖ_2 are

$$\varpi_1 = -\frac{(6y_0 y_1 + 2y_0^3 - 1)dy_0 + 2y_1 dy_1}{2 \det W}, \tag{4.25}$$

$$\varpi_2 = -dx - \frac{3\left((4y_1^2 - 4y_0^4 + 2y_0)dy_0 - 4\left(y_0 y_1 + y_0^3 - \frac{1}{2}\right) dy_1\right)}{4 \det W}. \tag{4.26}$$



Fig. 4.2 Sections of the graph of $u(t,x)$ for $t=0$ (red), 0.05 (blue), and 0.15 (green)

After integrating them, we get first integrals of the equation $F=0$. However, these integrals are cumbersome and we do not give them here. Fortunately, a general solution of equation

$$y'' + 3yy' + y^3 - \frac{1}{2} = 0 \tag{4.27}$$

can be constructed directly using Maple. Below we give the corresponding program code. As a result, we obtain the general solution of equation (4.27):

$$y(x) = \frac{C_1 e^{\frac{2}{\sqrt{3}}\chi} - \frac{1}{2}e^{-\frac{1}{\sqrt{3}}\chi}\left(\sqrt{3}C_2 + 1\right)\cos\,\chi + \frac{1}{2}e^{-\frac{1}{\sqrt{3}}\chi}\left(C_2 - \sqrt{3}\right)\sin\,\chi}{\sqrt[3]{2}\left(C_1 e^{\frac{2}{\sqrt{3}}\chi} - C_2 e^{-\frac{1}{\sqrt{3}}\chi}\sin\,\chi + e^{-\frac{1}{\sqrt{3}}\chi}\cos\,\chi\right)}, \tag{4.28}$$

where C_1 and C_2 are arbitrary constants and $\chi = \frac{\sqrt{3}}{\sqrt[3]{16}}x$.

An example of calculations in Maple is given below.

Maple Code: Second-order dynamics for the equation $u_t = u_{xx} - 2u^3 + 1$

1. Load libraries:

```
with(DifferentialGeometry):
with(JetCalculus):
```

```
with(Tools):
with(PDETools):
with(LinearAlgebra):
```

2. Set jet notation, declare coordinates on the manifold M, and generate coordinates on the 3-jet space:

```
Preferences("JetNotation", "JetNotation2"):
DGsetup([x], [y], M, 3):
```

3. Define the Poisson–Lie bracket on the space $J^3(\mathbb{R})$:

```
Poisson:= proc (A, B)
local i, P;
P:=0:
for i from 0 to 3 by 1 do
P:=P+(diff(A, y[i])*TotalDiff(B, [i])-
diff(B, y[i])*TotalDiff(A, [i]))
end do:
return P:
end proc:
```

4. Define the function ϕ and a second-order dynamics F:

```
f(y[0]) := -2*y[0]^3+1:
phi := y[2]+f(y[0]):
F:=y[2]-A(y[0])*y[1]-B(y[0]):
```

5. The Poisson–Lie bracket calculation:

```
eq0:=collect(Poisson(phi,F),{y[1],y[2]}):
```

6. Substitution of the second derivative:

```
sub_y2:=y[2]=solve(F,y[2]):
```

7. Restriction of the Poisson–Lie bracket to the dynamics $F = 0$:

```
eq1:=[coeffs(collect(eval(eq0,sub_y2),y[1]),y[1])]:
```

8. Printing the resulting equations $\overline{[\phi, F]} = 0$:

```
for i from 1 to nops(eq1) by 1 do
print(simplify(eq1[i]))
end do;
```

9. Solve the resulting system $\overline{[\phi, F]} = 0$ with respect to the functions A and B:

```
dsolve(eq1);
```

10. We get dynamics (4.18)–(4.20). Choose dynamics (4.18):

```
F:=y[2]+3y[0]*y[1]+y[0]^3-1/2:
```

11. Convert the function F to a differential operator (equation)

```
ode:=convert(F,DGdiff):
```

12. This equation can be solved by quadratures. Construct a solution of the Cauchy problem for this equation:

```
Y:=simplify(unapply(rhs(dsolve({ode,y(0)=a1,
(D(y))(0) = a2})),x,a1,a2)):
```

13. Restriction of ϕ to the dynamic $F = 0$:

```
sub_y2:=y[2]=solve(F,y[2]):
phi_F:=eval(phi, sub_y2):
```

14. Define vector field (4.15):

```
Z:=evalDG(D_x+y[1]*D_y[0]+(rhs(sub_y2))*D_y[1]):
```

15. Define vector field (4.23):

```
S1:=evalDG(phi_F*D_y[0]+LieDerivative(Z,phi_F)*D_y[1]):
```

16. To find the shifts Φ_t along the vector field S_1, we compose a system of differential equations

```
z1:=diff(q(t),t)=eval(Hook(S1,dy[0]),
  {y[0]=q(t),y[1]=p(t)});
z2:=diff(p(t),t)=eval(Hook(S1,dy[1]),
  {y[0]=q(t),y[1]=p(t)});
```

Here, $q = y_0$ and $p = y_1$.

17. Choose the solution of equation (4.27) with initial data

$$q(0) = a_0 = 4, \quad p(0) = a_1 = 2,$$

and compose a system to calculate the shift of the point (a_0, a_1):

```
ind:=q(0) = 4, p(0) = 2;
dsys:={z1,z2,ind};
```

18. Numerically solve this system:

```
dsn := dsolve(dsys, numeric);
```

19. Load the library:

```
with(plots):
```

20. Form the image of sections of the solution of the equation at $t = 0; 0.05; 0.15$:

```
r1:=plot(Y(x,rhs(dsn(0)[3]),rhs(dsn(0)[2])),
x=-3..4,color="RED");
r2:=plot(Y(x,rhs(dsn(0.05)[3]),rhs(dsn(0.1)[2])),
```

```
x=-3..4,color="Blue");
r3:=plot(Y(x,rhs(dsn(0.15)[3]),rhs(dsn(0.15)[2])),
x=-3..4,color="GREEN");
```

21. Display images on the screen:

```
display([r1,r2,r3],numpoints=1500,
resolution=3000,
thickness=2,axes = framed,
axesfont = ["TIMES", "ROMAN", 12],
labelfont = ["TIMES","ITALIC", 14],
labels = ["x", "y"],color="BLACK");
```

As a result, we obtain slices of the solution of the equation at moments $t = 0, 0.05, 0.15$ (see Fig. 4.2).

4.5 The Reaction–Diffusion Equation with a Convection Term

The reaction–diffusion equation with a nonlinear convection flow $H(u)$ in the positive direction of the x-axis has the form [14]

$$u_t + H(u)_x = u_{xx} + f(u). \tag{4.29}$$

Write this equation in the form

$$u_t = u_{xx} + g(u)u_x + f(u), \tag{4.30}$$

which is more convenient for calculations. Here, $g(u) = -H'(u)$. Then,

$$\phi(y_0, y_1, y_2) = y_2 + g(y_0)y_1 + f(y_0).$$

Below we suppose that

$$g' \neq 0. \tag{4.31}$$

4.5.1 *First-Order Dynamics*

Construct first-order dynamics of Eq. (4.30) in the following form:

$$F := y_1 - h(y_0) = 0, \tag{4.32}$$

where h is some smooth function.

The restriction of the Poisson–Lie bracket to Eq. (4.32) is

$$\overline{[\phi, F]} = h' f - h f' - h^2(g' + h'').$$

Equation $\overline{[\phi, F]} = 0$ has the trivial solution $h = 0$, which corresponds to x-independent solutions of equation (4.30). Consider the case when $h \neq 0$. Then, the function h satisfies the Abel differential equation of second kind (see [1])

$$hh' + (g(y_0) + \alpha)h + f(y_0) = 0, \tag{4.33}$$

where α is a constant. Due to (4.33), the evolutionary vector field has the form

$$S = \left(hh' + gh + f\right) \frac{\partial}{\partial y_0} = \alpha h \frac{\partial}{\partial y_0}.$$

4.5.2 Second-Order Dynamics

We will look for second-order dynamics in the form of the Liénard equation too (see (4.8)). The Poisson–Lie bracket is

$$[\phi, F] = -(g''+A'')y_1^3-(f''+B'')y_1^2-\left(2(g' + A')y_2 - g'B + A'f\right) y_1+B'f-f'B.$$

Since its restriction to Eq. (4.8) is a polynomial in y_1, Eq. (4.6) implies the following system of ordinary differential equation:

$$\begin{cases} (f - 2B)A' - 3g'B = 0, \\ 2(A' + g')A + f'' + B'' = 0, \\ B'f - Bf' = 0, \\ g'' + A'' = 0. \end{cases}$$

Solving this system, we find that the functions g and f should be linear and cubic, respectively:

$$f(y_0) = f_3 y_0^3 + f_2 y_0^2 + f_1 y_0 + f_0, \tag{4.34}$$

$$g(y_0) = g_1 y_0 + g_0, \tag{4.35}$$

where $f_0, \ldots, f_3, g_0, g_1 \in \mathbb{R}$. From inequality (4.31), it follows that $g_1 \neq 0$.

Theorem 4.3 *1. Equation (4.30) has second order finite dimensional dynamics in the form of the Liénard equation (4.8) if and only if the function f is a polynomial of third degree and the function g is linear.*

2. Suppose that the functions f and g have forms (4.34) and (4.35), respectively, where $f_3 \neq 0$, $g_1 \neq 0$. Then Eq. (4.30) has the finite dimensional dynamics

$$F = y_2 + (g_1 y_0 + \alpha) y_1 + f_3 y_0^3 + f_2 y_0^2 + f_1 y_0 + f_0,$$

where α is a constant. In addition, if the condition $g_1^2 - 8 f_3 \geq 0$ holds, then Eq. (4.30) has one more finite dimensional dynamics

$$\begin{aligned} F =& y_2 - (A_1 y_0 + A_0) y_1 + \frac{1}{3}(A_1^2 + g_1 A_1 + 3 f_3) y_0^3 \\ &+ (A_1 A_0 + f_2 + g_1 A_0) y_0^2 - B_1 y_0 - B_0, \end{aligned}$$

where

$$A_0 = \frac{f_2 \beta}{f_3}, \quad A_1 = 3\beta, \quad B_0 = \frac{f_0 (f_3 + g_1 \beta)}{2 f_3},$$

$$B_1 = \frac{f_1 (f_3 + g_1 \beta)}{2 f_3}, \quad \beta = \frac{-g_1 \pm \sqrt{g_1^2 - 8 f_3}}{4}.$$

Example 4.2 Consider the equation

$$u_t = u_{xx} - (u+1) u_x + \frac{1}{8} u^3. \tag{4.36}$$

Then,

$$\phi = y_2 - (y_0 + 1) y_1 + \frac{1}{8} y_0^3$$

and

$$F = y_2 - \frac{3}{4} y_0 y_1 + \frac{1}{16} y_0^3.$$

Restrict ϕ to the equation $F = 0$:

$$\overline{\phi} = -\frac{1}{4} y_0 y_1 + \frac{1}{16} y_0^3 - y_1.$$

The vector field is



Fig. 4.3 The graph of solution (4.38)

$$S=\left(-\frac{1}{4}y_0y_1+\frac{1}{16}y_0^3-y_1\right)\frac{\partial}{\partial y_0}+\left(-\frac{1}{4}y_1^2-\frac{3}{4}y_0y_1+\frac{1}{64}y_0^4+\frac{1}{16}y_0^3\right)\frac{\partial}{\partial y_1}.$$

The shift transformation Φ_t corresponding to this field is

$$x\longmapsto x,$$

$$y_0\longmapsto \frac{8(ty_0^2-4y_1t+4y_0)}{(t^2-2t)y_0^2+8ty_0+32+(8t-4t^2)y_1},$$

$$y_1\longmapsto \frac{8((2t+t^2)y_0^4+8ty_0^3-8ty_1(2+t)y_0^2-32ty_0y_1+128y_1+16t^2y_1^2+32ty_1^2)}{\left((t^2-2t)y_0^2+8ty_0-4t^2y_1+8y_1t+32\right)^2}.$$

The equation $F=0$ has the following general solution:

$$y(x)=-\frac{8(C_1x+C_2)}{C_1x^2+2C_2x+2}, \tag{4.37}$$

where C_1 and C_2 are arbitrary constants. Applying the inverse transformation Φ_t^{-1} to these functions, we get a solution of equation (4.36):

$$u(t,x)=-\frac{8(C_1(x-t)+C_2)}{2+((x-t)^2-2t)C_1+(2(x-t))C_2}. \tag{4.38}$$

The graph of this solution with $C_1 = C_2 = 1$ is shown in Fig. 4.3.

Below we give the part of the code responsible for the shift of solutions (4.37) along the trajectories of the vector field S.

Maple Code: Second-order dynamics for Eq. (4.36)

4. Define the function ϕ and a second-order dynamics F:

```
phi:=y[2]-(y[0]+1)*y[1]+(1/8)*y[0]^3:
F := y[2]-(3/4)*y[0]*y[1]+(1/16)*y[0]^3:
```

5. Restriction of ϕ to the dynamic $F = 0$:

```
phi_F:= phi-F:
```

6. Construct the vector field $\mathcal{D}$:

```
Z:=evalDG(D_x+y[1]*D_y[0]+(A(y[0])*y[1]+
B(y[0]))*D_y[1]):
```

7. Construct the vector field $\overline{S}$:

```
S:=evalDG(phi_F*D_y[0]+
LieDerivative(Z,phi_F)*D_y[1]):
```

8. The transformation Φ_t and its inverse transformation Φ_t^{-1}:

```
Phi:=Flow(S,t):
Xi:=InverseTransformation(Phi):
```

9. Solution (4.37) and its derivative:

```
ode:=convert(F1,DGdiff): v:=rhs(dsolve(ode));
w:=diff(v,x);
```

10. Apply the transformation Φ_t^{-1} to solution (4.37):

```
eq:=Pullback(Xi,[y[0]-v,y[1]-w]);
```

11. Find the explicit form of the solution:

```
sol:=solve(eq,{y[0],y[1]}):
q:=rhs(sol[1]);
```

12. Checking:

```
U:=diff(u(t,x),t)-eval(convert(phi,DGdiff),
y(x)=u(t,x));
simplify(eval(U,u(t,x)=q));
```

4.6 The Black–Scholes Equation

The Black–Scholes equation

$$u_t = -\frac{1}{2}\sigma^2 x^2 u_{xx} - rxu_x + ru \tag{4.39}$$

is a well-known linear partial differential equation of financial mathematics [4]. It describes the price of the option over time. Here, u is the price of the option as a function of stock price x and time t, r is the risk-free interest rate, and σ is the volatility of the stock. Note that, unlike the equations considered above, this equation depends on the variable x. For this equation,

$$\phi = -\frac{1}{2}\sigma^2 x^2 y_2 - rxy_1 + ry_0.$$

4.6.1 First-Order Dynamics

Since the equation is linear, we will seek its linear dynamics:

$$F = y_1 - A(x)y_0 - B(x). \tag{4.40}$$

Then,

$$[\phi, F] = \left(\frac{\sigma^2 x^2}{2} A'' y_0 + rxA'\right) y_0 + \left(r + \sigma^2 x^2 A'\right) y_1$$
$$+ \sigma^2 xy_2 + \frac{\sigma^2 x^2}{2} B'' + rxB' - rB.$$

The restriction $\overline{[\phi, F]}$ of this bracket in Eq. (4.40) is a linear function with respect to y_0. Therefore, the equation $\overline{[\phi, F]} = 0$ is equivalent to the following system of two ordinary differential equations:

$$\begin{cases} \frac{1}{2}A''\sigma^2 x^2 + (A\sigma^2 x + \sigma^2 + r)xA' + A^2\sigma^2 x + Ar = 0, \\ \frac{1}{2}B''\sigma^2 x + (\sigma^2 + r)B' + \sigma^2(xA' + A)B = 0. \end{cases}$$

The first equation can be solved:

$$A(x) = \frac{1}{2\sigma^2 x}\left(\sigma^2 - 2r - C_1 \tan\left(\frac{C_1 \ln x - C_2}{2\sigma^2}\right)\right),$$

where C_1 and C_2 are arbitrary constants. After substituting into the second equation, we obtain the equation for B:

$$4x\left(\frac{\sigma^2 x}{2}B'' + (\sigma^2 + r)B'\right)\sigma^2\cos^2\left(\frac{C_1 \ln x - C_2}{2\sigma^2}\right) - BC_1^2 = 0. \tag{4.41}$$

It is quite difficult to construct a general solution of this equation. But if we put $C_1 = 0$, then it can be easily solved:

$$B(x) = C_3 x^{-\frac{\sigma^2+2r}{\sigma^2}} + C_4,$$

where C_3 and C_4 are arbitrary constants. Then, we get the following dynamics:

$$F = y_1 - \frac{1}{2\sigma^2 x}(\sigma^2 - 2r)y_0 - C_3 x^{-\frac{\sigma^2+2r}{\sigma^2}} - C_4. \tag{4.42}$$

The general solution of the corresponding equation is

$$y(x) = C_5 x^{\frac{\sigma^2-2r}{2\sigma^2}} - \frac{2\sigma^2(C_3 x^{-\frac{2r}{\sigma^2}} - C_4 x)}{\sigma^2 + 2r}, \tag{4.43}$$

where C_5 is an arbitrary constant. A zero solution $B(x) = 0$ of Eq. (4.41) gives another dynamics:

$$F = y_1 - \frac{1}{2\sigma^2 x}\left(\sigma^2 - 2r - C_1 \tan\left(\frac{C_1 \ln x - C_2}{2\sigma^2}\right)\right)y_0. \tag{4.44}$$

Its general solution is

$$y(x) = C_3 x^{\frac{\sigma^2-2r}{2\sigma^2}} \cos\left(\frac{C_1 \ln x - C_2}{2\sigma^2}\right). \tag{4.45}$$

4.6.2 Construction Solutions of the Black–Scholes Equation by Dynamics

At first, consider dynamics (4.44). Restrict the function ϕ to dynamics (4.44):

$$\overline{\phi} = \frac{C_1^2 + (\sigma^2 + 2r)^2}{8\sigma^2}y_0.$$

Construct the vector field

$$\overline{S} = \overline{\phi}\frac{\partial}{\partial y_0},$$

and find its shift transformation:

$$\Phi_t : (x, y_0) \longmapsto \left(x, e^{\frac{C_1^2+(\sigma^2+2r)^2}{8\sigma^2}t} y_0 \right).$$

The inverse transformation is

$$\Phi_t^{-1} : (x, y_0) \longmapsto \left(x, e^{-\frac{C_1^2+(\sigma^2+2r)^2}{8\sigma^2}t} y_0 \right).$$

Acting by this transformation on function (4.45), we obtain the following exact solution of equation (4.39):

$$u(t, x) = C_3 x^{\frac{\sigma^2-2r}{2\sigma^2}} \cos\left(\frac{C_1 \ln x - C_2}{2\sigma^2} \right) e^{-\frac{C_1^2+(\sigma^2+2r)^2}{8\sigma^2}t}. \tag{4.46}$$

Here, C_1, C_2, and C_3 are arbitrary constants.

For example, the function

$$u(t, x) = \frac{e^{\frac{3}{2}t}}{\sqrt{2x}} \sqrt{\sin(\sqrt{3} \ln x) + 1} \tag{4.47}$$

is a solution of equation (4.39) with $\sigma = r = 1$ (see Figs. 4.4 and 4.5).

Now, consider dynamics (4.42). In this case,

$$\overline{\phi} = \frac{1}{8\sigma^2} \left(2r + \sigma^2 \right) \left((2r + \sigma^2) y_0 + 2\sigma^2 C_3 x^{-\frac{2r}{\sigma^2}} - 2C_4 \sigma^2 x \right).$$

Omitting cumbersome calculations, we write the final result. The function

$$u(t, x) = B_1 x + B_2 x^{-\frac{2r}{\sigma^2}} + B_3 x^{\frac{\sigma^2-2r}{2\sigma^2}} e^{\frac{(\sigma^2+2r)^2}{8\sigma^2}t}$$

is a solution of equation (4.39). Here, B_1, B_2, and B_3 are arbitrary constants.



Fig. 4.4 Sections of the graph of solution (4.47) for $t = 0$ (red), 0.1 (orange), 0.2 (green), and 0.3 (blue)



Fig. 4.5 The graph of solution (4.47)

Maple Code: First-order dynamics for the Black–Scholes equation

The first three items are the same as in the Maple code on page 135. We start at the fourth one.

4. Define the function ϕ and a second-order dynamics F:

```
phi := -sigma^2*x^2*y[2]/2-r*x*y[1]+r*y[0]:
F := y[1]-A(x)*y[0]-B(x):
```

5. The Poisson–Lie bracket calculation:

```
eq0:=simplify(Poisson(phi,F),size);
```

6. Substitution of the derivatives:

```
sub:=solve([F,TotalDiff(F,x)],{y[1],y[2]}):
```

7. Restriction of the Poisson–Lie bracket to the dynamics:

```
eq1:=simplify(eval(eq0,sub),size):
eq2:=[coeffs(collect(eq1,y[0],distributed),y[0])];
```

8. Print the resulting equations $\overline{[\phi, F]} = 0$:

```
for i from 1 to nops(eq2) by 1 do
print(simplify(eq2[i],size))
end do;
```

9. Next, the resulting system of equations is solved with respect to the function A and B in a semi-manual mode. As a result, we get dynamics (4.44):

```
F:=y[1]-(1/2)*(sigma^2-2*r
-tan((1/2)*C1*(ln(x)-C2)/sigma^2)*C1)*y[0]/
(sigma^2*x):
```

10. Restriction of ϕ to the dynamic $F = 0$:

```
phi_F:=simplify(eval(phi,sub),size):
```

11. Construct the vector field $\overline{S}$:

```
S:=evalDG(phi_F*D_y[0]):
```

12. The transformation Φ_t and its inverse transformation Φ_t^{-1}:

```
Phi:=Flow(S,t):
Xi:=InverseTransformation(Phi):
```

13. Apply the transformation Φ_t^{-1} to solution (4.45):

```
Xi_y:=Pullback(Xi,y[0]-(C3*x^((1/2)
*(sigma^2-2*r)/sigma^2)*cos((C1*ln(x)-C2)
/(2*sigma^2)))):
```

14. Find solution (4.46):

```
q:=solve(Xi_y,y[0]);
```

15. Check this solution:

```
BSch:=diff(u(t,x),t)-eval(convert(phi,DGdiff),
y(x)=u(t,x)):
simplify(eval(BSch,u(t,x)=q));
```

It should be zero.

Acknowledgments This work is partially supported by Russian Foundation for Basic Research (project 18-29-10013).

References

1. N. H. Abel, Précis d'une théorie des fonctions elliptiques, *J. Reine Angew. Math.*, 4(1829), 309.348.
2. Akhmetzyanov A. V., Kushner A. G., Lychagin V. V., Attractors in Models of Porous Media Flow, *Doklady. Mathematics* **472**:6 (2017), 627–630.
3. Salnikov A. M., Akhmetzianov A.V., Kushner A.G., Lychagin V.V., A Numerical Method for Constructing Attractors of Evolutionary Filtration Equations, *2019 1st International Conference on Control Systems, Mathematical Modelling, Automation and Energy Efficiency (SUMMA)*, Lipetsk, Russia (2019), pp. 22–24, https://doi.org/10.1109/SUMMA48161.2019.8947585.
4. Black F., Scholes M., The Pricing of Options and Corporate Liabilities, *The Journal of Political Economy* **81**:3 (1973), 637–654.
5. Duzhin S. V., Lychagin V. V., Symmetries of distributions and quadrature of ordinary differential equations, *Acta Appl. Math.* **24** (1991), 29–57.
6. Fisher R. A. The wave of advance of advantageous genes, *Annals of Eugenics.* **7** (1937), 353–369.
7. Kruglikov B. S., Lychagina O. V., Finite dimensional dynamics for Kolmogorov – Petrovsky – Piskunov equation, *Lobachevskii Journal of Mathematics* **19** (2005), 13–28.
8. A.N. Kolmogorov, I.G. Petrovskii, N.S. Piskunov, A study of diffusion with increase in the quantity of matter, and its application to a biological problem, *Bull. Moscow State Univ.* **17** (1937), 1–72.
9. Krasilshchik I. S., Lychagin V. V., Vinogradov A. M., Geometry of jet spaces and nonlinear partial differential equations, New York: Gordon and Breach, 1986.
10. Kushner A. G., Matviichuk R.I., Exact solutions of the Burgers – Huxley equation via dynamics, *Journal of Geometry and Physics **151** (2020)*
11. Kushner A. G., Lychagin V. V., Rubtsov V. N., Contact geometry and nonlinear differential equations, *Encyclopedia of Mathematics and Its Applications,* **101**. Cambridge: Cambridge University Press, xxii+496 pp., 2007.
12. Liénard A., Etude des oscillations entretenues, *Revue générale de l'électricité* **23** (1928), 901–912 and 946–954.
13. Lychagin V. V., Lychagina O. V., Finite Dimensional Dynamics for Evolutionary Equations, *Nonlinear Dyn.*, **48** (2007), 29–48.
14. Murray J. D., Mathematical Biology, Springer-Verlag, New York 1993.

15. NagumoJ., Yoshizawa S., and Arimoto S., Bistable transmission lines, *Circuit Theory, IEEE Transactions on*, **12**,no 3 (1965), 400–412.
16. Nourazar S.S., Soori M. and Nazari-Golshan A., On the Exact Solution of Burgers – Huxley Equation Using the Homotopy Perturbation Method, *Journal of Applied Mathematics and Physics* **3** (2015), 285–294.
17. Osipov V.V., The simplest autowaves, *Soros Educational Journal*, **7** (1999), 115–121.
18. Satsuma J. Exact solutions of Burgers equation with reaction terms. In: Topics in soliton theory and exactly solvable nonlinear equations. (M.J. Ablowitz, B. Fuchssteiner, M.D. Kruskal eds). Singapore: World Scientific, 1987. pp. 255–262.
19. Singh B., Kumar R., Exact Solutions of Certain Nonlinear Diffusion-Reaction Equations with a Nonlinear Convective Term, *International Journal of Pure and Applied Physics* **13**:1 (2017), 125–132.
20. Tychkov S.N. Introduction to Symbolic Computations in Differential Geometry with Maple, *Nonlinear PDEs, Their Geometry, and Applications*, Birkhäuser, Cham (2019). 157–182.
21. Zeldovich Y., Frank-Kamenetskii D.A., A theory of thermal propagation of flame, *Acta Physicochimica USSR*, **9** (1938) 341–350.